
'--------Title--------

' File......read_voltage.pbp
' Started....5/28/12
' Microcontroller used:  Microchip Technology PIC16F88
'                        microchip.com
' PicBasic Pro Code: micro-Engineering Labs, Inc.
'                    melabs.com 

'--------Program Desciption--------

' Prints 0 to +15 volts reading to 16 x 2 parallel 
' LCD which uses Hitachi 44780 controller.

'----------Related Lessons---------

' See LCD BASICS lesson at: 
' http://cornerstonerobotics.org/curriculum/lessons_year2/erii14_lcd1.pdf
'
' read_voltage.pbp is used in the lesson Control and Navigation 4 at:
' 

'------------Comments--------------

'-----New PicBasic Pro Command-----

' The PicBasic Pro Compiler Manual is on line at:
' http://www.microengineeringlabs.com/resources/index.htm#Manuals 

'---------PIC Connections---------

'       16F88 Pin           Wiring
'       ---------           ---------- 
'       RA0                 LCD pin 11(DB4)
'       RA1                 LCD pin 12(DB5)
'       RA2                 LCD pin 13(DB6)
'       RA3                 LCD pin 14(DB7)
'       RA4                 LCD Register Select(RS)
'       RB3                 LCD Enable(E)
'       RB7                 Voltage Divider Network
'       See schematic for the other usual PIC connections

'---------LCD Connections---------

'        LCD Pin              Wiring
'       ---------           ---------- 
'          1               Ground(Vss)
'          2               + 5v(Vdd)
'          3               Center of 20K Pot(Contrast)
'          4               RA4(Register Select,RS)
'          5               Ground(Read/Write,R/W)
'          6               RB3(Enable)
'          7               No Connection(DB0) 
'          8               No Connection(DB1)     
'          9               No Connection(DB2)     

'         10               No Connection(DB3)     
'         11               RA0(DB4)    
'         12               RA1(DB5)
'         13               RA2(DB6)
'         14               RA3(DB7)

'---------Revision History--------

'----------Initialization---------

    TRISB = %10000000          ' Set Pin RB7 as an input
    
    ANSEL = %01000000          ' Set channel 6 as analog and the other
                               ' channels as digital.
                                
    OSCCON = $60               ' Sets the internal oscillator in the
                               ' 16F88 to 4 MHz 
                                        
    x           VAR     BYTE   ' Defines x as a BYTE variable
    y           VAR     WORD   ' Defines y as a WORD varisable 
    VHigh       VAR     BYTE   ' Defines VHigh as a Byte variable
    VLow        VAR     BYTE   ' Defines VLow as a Byte variable
    
'------------Main Code------------

    PAUSE 1000                 ' Pause to allow LCD to setup
    
DO                             ' DO loop
    ADCIN 6, x                 ' Read analog voltage on AN6 
                               ' (pin RB7) and convert to 8-bit 
                               ' digital value then store as x.
    
    y = (x*30)/51              ' Multiply 8-bit digital value x by 30
                               ' and then divide by 51 to get voltage
                               ' reading times 10.
                               ' Example: If 8-bit digital value = 255,
                               ' (255*30)/51 = 150 (10 times the voltage
                               ' reading of 15.0 volts.    
    
    VHigh = y/10               ' Divide y by 10 to get the whole
                               ' number digits of the voltage reading.
                               ' Place result in VHigh.
                               ' Example: 150/10 = 15
                                                        
    
    VLow = y//10               ' '//' operator returns the remainder.
                               ' y//10 divides y by 10 and places the
                               ' remainder in VLow.
                               ' Example: 150//10 = 0
                               ' 150/10 = 15.0
                               ' The remainder 0 is placed in VLow.   
                               
        
    LCDOUT $FE,1,"Voltage = ", #VHigh, ".", #VLow       
                               ' Prints the whole number digits of the 

                               ' voltage, VHigh, then a decimal point ".",
                               ' then the decimal digit VLow.
    
    PAUSE 1000                 ' 1 second pause
    
LOOP                                                  
    
END    

